Resonant tunnelling example.

Resonant tunnelling from a piecewise flat potential:

V(x) = 0, |x|>b and |x|<a,

= v, a<|x|<b,

with b>a>0. Define

$$k = \sqrt{2 m E} \Big/ \hbar,$$

$$\kappa = \sqrt{2 m (v - E)} \Big/ \hbar.$$

Define the matrices for matching conditions at x=±a, ±b, are

```
M1[a_] = {{e^(i k a), e^(-i k a)}, {i k e^(i k a), -i k e^(-i k a)}};
M2[a_] = {{e^(κ a), e^(-κ a)}, {κ e^(κ a), -κ e^(-κ a)}};
M1[x] // MatrixForm
M2[x] // MatrixForm
```

$$\begin{pmatrix} e^{i k x} & e^{-i k x} \\ i e^{i k x} k & -i e^{-i k x} k \end{pmatrix}$$

$$\begin{pmatrix} e^{x \kappa} & e^{-x \kappa} \\ e^{x \kappa} \kappa & -e^{-x \kappa} \kappa \end{pmatrix}$$

Then, if the wave function for |x|>b has the form

ψ(x) = A $e^{i k x}$ + B $e^{-i k x}$, for x < -b,

= C $e^{i k x}$, for x > b,

the matching conditions can be written as the matrix equation $\vec{a} = X \vec{c}$ where $\vec{a}^T = (A, B)$, $\vec{c}^T = (C, 0)$, and

```
X = Simplify[
   Inverse[M1[-b]].M2[-b].Inverse[M2[-a]].M1[-a].Inverse[M1[a]].M2[a].Inverse[M2[b]].M1[b]];
```

Then C/A = $1/X_{11}$ where

```
X11 = X[[1, 1]]
```

$$\frac{1}{16 k^2 \kappa^2} e^{-2 i a k + 2 i b k - 2 a \kappa - 2 b \kappa} \Big(-e^{4 b \kappa} (k + i \kappa)^4 - e^{4 a \kappa} (i k + \kappa)^4 + 2 e^{2 (a+b) \kappa} \left(k^2 + \kappa^2\right)^2 + e^{4 a (i k + \kappa)} \left(k^2 + \kappa^2\right)^2 - 2 e^{4 i a k + 2 a \kappa + 2 b \kappa} \left(k^2 + \kappa^2\right)^2 + e^{4 i a k + 4 b \kappa} \left(k^2 + \kappa^2\right)^2\Big)$$

so the transmission probability is given by T = $1/|X_{11}|^2$:

```
Xexp = ComplexExpand[X11];
Xconj = Xexp /. ⅈ → -ⅈ;
Xsquared = Simplify[Expand[Xexp Xconj]]
```

$$-\frac{1}{128\, k^4 \kappa^4} e^{-4 (a+b) \kappa} \Big(-e^{8 a \kappa} k^8 - e^{8 b \kappa} k^8 - 6 e^{4 (a+b) \kappa} k^8 + 4 e^{2 (3 a+b) \kappa} k^8 + 4 e^{2 (a+3 b) \kappa} k^8 - 4 e^{8 a \kappa} k^6 \kappa^2 - 4 e^{8 b \kappa} k^6 \kappa^2 + 8 e^{4 (a+b) \kappa} k^6 \kappa^2 - 6 e^{8 a \kappa} k^4 \kappa^4 - 6 e^{8 b \kappa} k^4 \kappa^4 - 100 e^{4 (a+b) \kappa} k^4 \kappa^4 - 8 e^{2 (3 a+b) \kappa} k^4 \kappa^4 - 8 e^{2 (a+3 b) \kappa} k^4 \kappa^4 - 4 e^{8 a \kappa} k^2 \kappa^6 - 4 e^{8 b \kappa} k^2 \kappa^6 + 8 e^{4 (a+b) \kappa} k^2 \kappa^6 - e^{8 a \kappa} \kappa^8 - e^{8 b \kappa} \kappa^8 - 6 e^{4 (a+b) \kappa} \kappa^8 + 4 e^{2 (3 a+b) \kappa} \kappa^8 + 4 e^{2 (a+3 b) \kappa} \kappa^8 + \left(e^{2 a \kappa} - e^{2 b \kappa}\right)^2 \left(k^2+\kappa^2\right)^2 \left(-2 e^{2 (a+b) \kappa} \left(k^2+\kappa^2\right)^2 + e^{4 a \kappa} \left(k^4 - 6 k^2 \kappa^2 + \kappa^4\right) + e^{4 b \kappa} \left(k^4 - 6 k^2 \kappa^2 + \kappa^4\right)\right) \mathrm{Cos}[4 a k] + 4 \left(e^{2 a \kappa} - e^{2 b \kappa}\right)^3 \left(e^{2 a \kappa} + e^{2 b \kappa}\right) k \kappa \left(-k^2+\kappa^2\right) \left(k^2+\kappa^2\right)^2 \mathrm{Sin}[4 a k]\Big)$$

This is a mess, but plot it for some convenient values (a=$\pi/2$, b=π, $E = n^2$, v=100, m/(2$\hbar$)=1), so that the two barriers each have thickness $\pi/2$:

```
WellTransmission = 1 / Xsquared /. {a → π / 2, b → π, k → n, κ → √(100 - n²)};
welltranplot = Plot[Log[WellTransmission], {n, 0, 10}, PlotRange → All, PlotPoints → 100 000]
```



Note the peaks in T at n ≈ {1,2,3,4,...}. These values of n correspond to the bound state energies of the square well potential in the middle (x<|a|). These peaks in the transmission probability are called "resonant tunnelling".

For comparison, consider instead the transmission probability through a single square barrier of thickness π (i.e., without an intervening potential well). Following the same steps as above gives ...

```
B = Simplify[Inverse[M1[-b]].M2[-b].Inverse[M2[-a]].M1[-a]];
B11 = B[[1, 1]];
Bexp = ComplexExpand[B11];
Bconj = Bexp /. ⅈ → -ⅈ;
Bsquared = Simplify[Expand[Bexp Bconj]]
BarrierTransmission = 1 / Bsquared /. {a → 0, b → π, k → n, κ → Sqrt[100 - n^2]};
Plot[{Log[WellTransmission], Log[BarrierTransmission]},
 {n, 0, 10}, PlotRange → All, PlotPoints → 100 000]
```

$$\frac{e^{-2(a+b)\kappa}\left(e^{4a\kappa}\left(k^2+\kappa^2\right)^2+e^{4b\kappa}\left(k^2+\kappa^2\right)^2-2e^{2(a+b)\kappa}\left(k^4-6k^2\kappa^2+\kappa^4\right)\right)}{16k^2\kappa^2}$$



... which closely follows the transmission probability for the case with an intervening potential well, except for the absence of the resonant tunnelling peaks.

[Note: In the plot the resonant tunnelling peaks only seem to rise about 10 orders of magnitude (i.e., a factor of about e^{25}) above the background, but this is just because of the numerical resolution of *Mathematica* --- in fact, they rise to values much closer to 1 (i.e., log(T) ≈ 0). Numerical evidence of this is just to evaluate the peak values by searching for relevant extrema of T. This gives the points in the following plot...

```
Off[FindRoot::"lstol"]
Off[Power::"infy"]
peak[1] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, .95}];
val[1] = {y → Log[Abs[WellTransmission /. peak[1]]]};
peak[2] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 1.9}];
val[2] = {y → Log[Abs[WellTransmission /. peak[2]]]};
peak[3] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 2.8}];
val[3] = {y → Log[Abs[WellTransmission /. peak[3]]]};
peak[4] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 3.75}];
val[4] = {y → Log[Abs[WellTransmission /. peak[4]]]};
peak[5] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 4.7}];
val[5] = {y → Log[Abs[WellTransmission /. peak[5]]]};
peak[6] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 5.6}];
val[6] = {y → Log[Abs[WellTransmission /. peak[6]]]};
peak[7] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 6.55}];
val[7] = {y → Min[Log[Abs[WellTransmission /. peak[7]]], 0]};
peak[8] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 7.4}];
val[8] = {y → Min[Log[Abs[WellTransmission /. peak[8]]], 0]};
peak[9] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 8.35}];
val[9] = {y → Log[Abs[WellTransmission /. peak[9]]]};
peak[10] = FindRoot[D[Denominator[WellTransmission], n] == 0, {n, 9.24}];
val[10] = {y → Log[Abs[WellTransmission /. peak[10]]]};
peaks = Table[{n /. peak[i], y /. val[i]}, {i, 1, 10}];
peakplot = ListPlot[peaks];
Show[welltranplot, peakplot]
```



... showing another factor of e^{10} increase in T over the line plot. Also, the fact that for the last 4 peaks the value is essentially T=1 (log(T)=0), indicates that the values at the earlier peaks are probably just limited by numerical accuracy.]